
*=============One_OPAmp_Current_Source============
One common application for a Class D output stage is
to drive LEDs. But using Class D output stages that use
feedback create some stability challenges. The
following example shows how feedback is used to create a
one Op Amp current source. 

=======================================================================
One_OpAmp_Current_Source
*
*                     R1A      INN        R1B
*     _____________/\  /\  /\___________/\  /\  /\________
*   _|_              \/  \/    |          \/  \/          |
*  /V1 \                       | |\                       |
*  \___/                       |_|-\                  V4  |
*   _|_                          |  \_____________________|
*   ///                     INP  |  /                     |
*                              __|+/                      |
*               V3             | |/ OPA                   |
*         R1C         R3A      |            VOUT    R3B   |
*     _/\  /\  /\__/\  /\  /\__|__/\  /\  /\___/\  /\  /\_|
*     |  \/  \/      \/  \/      R1D\/  \/   |   \/  \/
*     |      VC1         VCB                 |
*    _|_                ______               |
*   /V2 \             _|_     |              |
*   \___/            /_  \    |          RL  |
*    _|_            // \  \VCM|___/\  /\  /\_|
*    ///            \   \//         \/  \/
*                    \___/
*                     _|_
*                     ///
*
*.OPTIONS GMIN=1f       METHOD=trap   ABSTOL=1u     TEMP=27   srcsteps = 1  gminsteps = 1 
*.OPTIONS RELTOL=.001   ABSTOL=1n     VNTOL=1u      ITL4=500  ITL1=400  
*=========Create_Signal==================
VCM       VCM    0      DC     0      SIN(   0     .25    1k      )
V1        V1     0      DC     1
V2        V2     0      DC     2
B_OPA     V4     0      V =    5*tanh(tanh((V(INP)-V(INN))*30)*30)
R1A       V1     INN    1K
R1B       INN    V4     1K
R1C       V2     V3     1K
R1D       INP    VOUT   1K
R3A       V3     INP    2K
R3B       VOUT   V4     2K
RL        VOUT   VCM    1K
.control
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran      .1u    3m     0      .1u   
set       pensize = 2
plot      vout VCM  vout-vcm
.endc
.end

=======================================================================
The Op Amp is simulated using the tanH function. 
Applying an AC sine-wave to voltage source VCM shows
that the feedback from VOUT will hold the current 
through RL constant.  

=======================================================================



This circuit was stolen from Bob Pease. Thank Bob!



===========Full_Netlist_For_Copy_Paste=======================
One_OpAmp_Current_Source
*
*                     R1A      INN        R1B
*     _____________/\  /\  /\___________/\  /\  /\________
*   _|_              \/  \/    |          \/  \/          |
*  /V1 \                       | |\                       |
*  \___/                       |_|-\                  V4  |
*   _|_                          |  \_____________________|
*   ///                     INP  |  /                     |
*                              __|+/                      |
*               V3             | |/ OPA                   |
*         R1C         R3A      |            VOUT    R3B   |
*     _/\  /\  /\__/\  /\  /\__|__/\  /\  /\___/\  /\  /\_|
*     |  \/  \/      \/  \/      R1D\/  \/   |   \/  \/
*     |      VC1         VCB                 |
*    _|_                ______               |
*   /V2 \             _|_     |              |
*   \___/            /_  \    |          RL  |
*    _|_            // \  \VCM|___/\  /\  /\_|
*    ///            \   \//         \/  \/
*                    \___/
*                     _|_
*                     ///
*
*.OPTIONS GMIN=1f       METHOD=trap   ABSTOL=1u     TEMP=27   srcsteps = 1  gminsteps = 1 
*.OPTIONS RELTOL=.001   ABSTOL=1n     VNTOL=1u      ITL4=500  ITL1=400  
*=========Create_Signal==================
VCM       VCM    0      DC     0      SIN(   0     .25    1k      )
V1        V1     0      DC     1
V2        V2     0      DC     2
B_OPA     V4     0      V =    5*tanh(tanh((V(INP)-V(INN))*30)*30)
R1A       V1     INN    1K
R1B       INN    V4     1K
R1C       V2     V3     1K
R1D       INP    VOUT   1K
R3A       V3     INP    2K
R3B       VOUT   V4     2K
RL        VOUT   VCM    1K

.control
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran      .1u    3m     0      .1u   
set       pensize = 2
plot      vout VCM  vout-vcm
.endc

.end

